

biplot — Biplots[Description](#)[Quick start](#)[Menu](#)[Syntax](#)[Options](#)[Remarks and examples](#)[Stored results](#)[Methods and formulas](#)[Acknowledgment](#)[References](#)[Also see](#)

Description

`biplot` displays a two-dimensional biplot of a dataset. A biplot simultaneously displays the observations (rows) and the relative positions of the variables (columns). Marker symbols (points) are displayed for observations, and arrows are displayed for variables. Observations are projected to two dimensions such that the distance between the observations is approximately preserved. The cosine of the angle between arrows approximates the correlation between the variables.

Quick start

Biplot of `v1`, `v2`, `v3`, and `v4`

```
biplot v1 v2 v3 v4
```

Same as above, but use standardized instead of centered variables

```
biplot v1 v2 v3 v4, std
```

Same as above, but graph dimension 3 versus 1 instead of 2 versus 1

```
biplot v1 v2 v3 v4, std dim(3 1)
```

Biplot with separate row and column plots

```
biplot v1 v2 v3 v4, separate
```

Only graph the column plot

```
biplot v1 v2 v3 v4, norow
```

Biplot with different color row markers for each category of `catvar`

```
biplot v1 v2 v3 v4, rowover(catvar)
```

Label row observations using the values in variable `mylabel`

```
biplot v1 v2 v3 v4, rowlabel(mylabel)
```

Store biplot coordinates in new variables `x` and `y`

```
biplot v1 v2 v3 v4, generate(x y)
```

Menu

Statistics > Multivariate analysis > Biplot

Syntax

```
biplot varlist [if] [in] [, options]
```

<i>options</i>	Description
Main	
<code>rowover(<i>varlist</i>)</code>	identify observations from different groups of <i>varlist</i> ; may not be combined with <code>separate</code> or <code>norow</code>
<code>dim(# #)</code>	two dimensions to be displayed; default is <code>dim(2 1)</code>
<code>std</code>	use standardized instead of centered variables
<code>alpha(#)</code>	row weight = #; column weight = 1 - #; default is 0.5
<code>stretch(#)</code>	stretch the column (variable) arrows
<code>mahalanobis</code>	approximate Mahalanobis distance; implies <code>alpha(0)</code>
<code>xnegate</code>	negate the data relative to the <i>x</i> axis
<code>ynegate</code>	negate the data relative to the <i>y</i> axis
<code>autoaspect</code>	adjust aspect ratio on the basis of the data; default aspect ratio is 1
<code>separate</code>	produce separate plots for rows and columns; may not be combined with <code>rowover()</code>
<code>nograph</code>	suppress graph
<code>table</code>	display table showing biplot coordinates
Rows	
<code>rowopts(<i>row_options</i>)</code>	affect rendition of rows (observations)
<code>row#opts(<i>row_options</i>)</code>	affect rendition of rows (observations) in the #th group of <i>varlist</i> defined in <code>rowover()</code> ; available only with <code>rowover()</code>
<code>rowlabel(<i>varname</i>)</code>	specify label variable for rows (observations)
<code>norow</code>	suppress row points; may not be combined with <code>rowover()</code>
<code>generate(<i>newvar_x</i> <i>newvar_y</i>)</code>	store biplot coordinates for observations in variables <i>newvar_x</i> and <i>newvar_y</i>
Columns	
<code>colopts(<i>col_options</i>)</code>	affect rendition of columns (variables)
<code>negcol</code>	include negative column (variable) arrows
<code>negcolopts(<i>col_options</i>)</code>	affect rendition of negative columns (variables)
<code>nocolumn</code>	suppress column arrows
Y axis, X axis, Titles, Legend, Overall	
<code>twoway_options</code>	any options other than <code>by()</code> documented in [G-3] <i>twoway_options</i>
collect is allowed; see [U] 11.1.10 Prefix commands.	
row_options	
Description	
<code>marker_options</code>	change look of markers (color, size, etc.)
<code>marker_label_options</code>	change look or position of marker labels
<code>no-label</code>	remove the default row (variable) label from the graph
<code>name(<i>name</i>)</code>	override the default name given to rows (observations)

<i>col_options</i>	Description
<i>pcarrow_options</i>	affect the rendition of paired-coordinate arrows
<code>no label</code>	remove the default column (variable) label from the graph
<code>name(<i>name</i>)</code>	override the default name given to columns (variables)

Options

Main

`rowover(varlist)` distinguishes groups among observations (rows) by highlighting observations on the plot for each group identified by equal values of the variables in *varlist*. By default, the graph contains a legend that consists of group names. `rowover()` may not be combined with `separate` or `norow`.

`dim(# #)` identifies the dimensions to be displayed. For instance, `dim(3 2)` plots the third dimension (vertically) versus the second dimension (horizontally). The dimension numbers cannot exceed the number of variables. The default is `dim(2 1)`.

`std` produces a biplot of the standardized variables instead of the centered variables.

`alpha(#)` specifies that the variables be scaled by $\lambda^\#$ and the observations by $\lambda^{(1-\#)}$, where λ are the singular values. It is required that $0 \leq \# \leq 1$. The most common values are 0, 0.5, and 1. The default is `alpha(0.5)` and is known as the symmetrically scaled biplot or symmetric factorization biplot. The result with `alpha(1)` is the principal-component biplot, also called the row-preserving metric (RPM) biplot. The biplot with `alpha(0)` is referred to as the column-preserving metric (CPM) biplot.

`stretch(#)` causes the length of the arrows to be multiplied by $\#$. For example, `stretch(1)` would leave the arrows the same length, `stretch(2)` would double their length, and `stretch(0.5)` would halve their length.

`mahalanobis` implies `alpha(0)` and scales the positioning of points (observations) by $\sqrt{n-1}$ and positioning of arrows (variables) by $1/\sqrt{n-1}$. This additional scaling causes the distances between observations to change from being approximately proportional to the Mahalanobis distance to instead being approximately equal to the Mahalanobis distance. Also, the inner products between variables approximate their covariance.

`xnegate` specifies that dimension-1 (*x* axis) values be negated (multiplied by -1).

`ynegate` specifies that dimension-2 (*y* axis) values be negated (multiplied by -1).

`autoaspect` specifies that the aspect ratio be automatically adjusted based on the range of the data to be plotted. This option can make some biplots more readable. By default, `biplot` uses an aspect ratio of one, producing a square plot. Some biplots will have little variation in the *y*-axis direction, and using the `autoaspect` option will better fill the available graph space while preserving the equivalence of distance in the *x* and *y* axes.

As an alternative to `autoaspect`, the *twoway_option* `aspectratio()` can be used to override the default aspect ratio. `biplot` accepts the `aspectratio()` option as a suggestion only and will override it when necessary to produce plots with balanced axes; that is, distance on the *x* axis equals distance on the *y* axis.

twoway_options, such as `xlabel()`, `xscale()`, `ylabel()`, and `yscale()`, should be used with caution. These *axis_options* are accepted but may have unintended side effects on the aspect ratio. See [G-3] *twoway_options*.

`separate` produces separate plots for the row and column categories. The default is to overlay the plots. `separate` may not be combined with `rowover()`.

`nograph` suppresses displaying the graph.

`table` displays a table with the biplot coordinates.

Rows

`rowopts(row_options)` affects the rendition of the points plotting the rows (observations). This option may not be combined with `rowover()`. The following *row_options* are allowed:

marker_options affect the rendition of markers drawn at the plotted points, including their shape, size, color, and outline; see [G-3] [marker_options](#).

marker_label_options specify the properties of marker labels; see [G-3] [marker_label_options](#). `mlabel()` in `rowopts()` may not be combined with the `rowlabel()` option.

`no-label` removes the default row label from the graph.

`name(name)` overrides the default name given to rows.

`row#opts(row_options)` affects rendition of the points plotting the rows (observations) in the #th group identified by equal values of the variables in *varlist* defined in `rowover()`. This option requires specifying `rowover()`. See `rowopts()` above for the allowed *row_options*, except `mlabel()` is not allowed with `row#opts()`.

`rowlabel(varname)` specifies label variable for rows (observations).

`norow` suppresses plotting of row points. This option may not be combined with `rowover()`.

`generate(newvarx newvary)` stores biplot coordinates for rows in variables *newvar_x* and *newvar_y*.

Columns

`colopts(col_options)` affects the rendition of the arrows and points plotting the columns (variables). The following *col_options* are allowed:

pcarrow_options affect the rendition of paired-coordinate arrows; see [G-2] [graph twoway pcarrow](#).

`no-label` removes the default column label from the graph.

`name(name)` overrides the default name given to columns.

`negcol` includes negative column (variable) arrows on the plot.

`negcolopts(col_options)` affects the rendition of the arrows and points plotting the negative columns (variables). The *col_options* allowed are given [above](#).

`nocolumn` suppresses plotting of column arrows.

Y axis, X axis, Titles, Legend, Overall

twoway_options are any of the options documented in [G-3] [twoway_options](#), excluding `by()`. These include options for titling the graph (see [G-3] [title_options](#)) and for saving the graph to disk (see [G-3] [saving_option](#)). See [autoaspect](#) above for a warning against using options such as `xlabel()`, `xscale()`, `ylabel()`, and `yscale()`.

Remarks and examples

The `biplot` command produces what [Cox and Cox \(2001\)](#) refer to as the “classic biplot”. Biplots were introduced by [Gabriel \(1971\)](#); also see [Gabriel \(1981\)](#). [Gower and Hand \(1996\)](#) discuss extensions and generalizations to biplots and place many of the well-known multivariate techniques into a generalized biplot framework extending beyond the classic biplot implemented by Stata’s `biplot` command. [Cox and Cox \(2001\)](#), [Jolliffe \(2002\)](#), [Gordon \(1999\)](#), [Jacoby \(1998\)](#), [Rencher and Christensen \(2012\)](#), and [Seber \(1984\)](#) discuss the classic biplot. [Kohler \(2004\)](#) provides a Stata implementation of biplots.

Let \mathbf{X} be the centered (or standardized if the `std` option is specified) data. A biplot splits the information in \mathbf{X} into a portion related to the observations (rows of \mathbf{X}) and a portion related to the variables (columns of \mathbf{X})

$$\mathbf{X} \approx (\mathbf{U}_2 \mathbf{\Lambda}_2^\alpha)(\mathbf{V}_2 \mathbf{\Lambda}_2^{1-\alpha})'$$

where $0 \leq \alpha \leq 1$; see [Methods and formulas](#) for details. $\mathbf{U}_2 \mathbf{\Lambda}_2^\alpha$ contains the plotting coordinates corresponding to observations (rows), and $\mathbf{V}_2 \mathbf{\Lambda}_2^{1-\alpha}$ contains the plotting coordinates corresponding to variables (columns). In a biplot, the row coordinates are plotted as symbols, and the column coordinates are plotted as arrows from the origin.

The commonly used values for α are 0, 0.5, and 1. The default is 0.5. The `alpha()` option allows you to set α .

Biplots with an α of 1 are also called principal-component biplots because $\mathbf{U}_2 \mathbf{\Lambda}_2$ contains the principal-component scores and \mathbf{V}_2 contains the principal-component coefficients. Euclidean distance between points in this kind of biplot approximates the Euclidean distance between points in the original higher-dimensional space.

Using an α of 0, Euclidean distances in the biplot are approximately proportional to Mahalanobis distances in the original higher-dimensional space. Also, the inner product of the arrows is approximately proportional to the covariances between the variables.

When you set α to 0 and specify the `mahalanobis` option, the Euclidean distances are not just approximately proportional but are approximately equal to Mahalanobis distances in the original space. Likewise, the inner products of the arrows are approximately equal (not just proportional) to the covariances between the variables. This means that the length of an arrow is approximately equal to the standard deviation of the variable it represents. Also, the cosine of the angle between two arrows is approximately equal to the correlation between the two variables.

A biplot with an α of 0.5 is called a symmetric factorization biplot or symmetrically scaled biplot. It often produces reasonable looking biplots where the points corresponding to observations and the arrows corresponding to variables are given equal weight. Using an α of 0 (or 1) causes the points (or the arrows) to be bunched tightly around the origin while the arrows (or the points) are predominant in the graph. Here many authors recommend picking a scaling factor for the arrows to bring them back into balance. The `stretch()` option allows you to do this.


Regardless of your choice of α , the position of a point in relation to an arrow indicates whether that observation is relatively large, medium, or small for that variable. Also, although the special conditions mentioned earlier may not strictly hold for all α , the biplot still aids in understanding the relationship between the variables, the observations, and the observations and variables jointly.

Example 1

Gordon (1999, 176) provides a simple example of a biplot based on data having five rows and three columns.

```
. input v1 v2 v3
      v1      v2      v3
1.   60   80 -240
2.  -213  66  180
3.   123 -186  180
4.    -9  38  -60
5.   39   2  -60
6. end


. biplot v1 v2 v3
Biplot of 5 observations and 3 variables
  Explained variance by component 1 = 0.6283
  Explained variance by component 2 = 0.3717
  Total explained variance = 1.0000
```



The first component accounts for 63% of the variance, and the second component accounts for the remaining 37%. All the variance is accounted for because, here, the 5-by-3 data matrix is only of rank 2.

Gordon actually used an α of 0 and performed the scaling to better match Mahalanobis distance. We do the same using the options `alpha(0)` and `mahalanobis`. (We could just use `mahalanobis` because it implies `alpha(0)`.) With an α of 0, Gordon decided to scale the arrows by a factor of 0.01. We accomplish this with the `stretch(.01)` option and add options to provide a title and subtitle in place of the default title obtained previously.

```
. biplot v1 v2 v3, alpha(0) mahalanobis stretch(.01) title(Simple biplot)
> subtitle(See figure 6.10 of Gordon (1999))
Biplot of 5 observations and 3 variables
  Explained variance by component 1 = 0.6283
  Explained variance by component 2 = 0.3717
  Total explained variance = 1.0000
```



The outcome is little changed between the first and second biplot except for the additional titles and the scale of the x and y axes.

`biplot` allows you to highlight observations belonging to different groups by using option `rowover()`. Suppose our data come from two groups defined by variable `group`, `group=1` and `group=2`.


```
. generate byte group = cond(_n<3, 1, 2)
. list
```

	v1	v2	v3	group
1.	60	80	-240	1
2.	-213	66	180	1
3.	123	-186	180	2
4.	-9	38	-60	2
5.	39	2	-60	2

Here is the previous biplot with group-specific markers:

```
. biplot v1 v2 v3, alpha(0) mahalanobis stretch(.01) title(Simple biplot)
> subtitle(Grouping variable group) rowover(group)
> row1opts(name("Group 1") msymbol(0) nolabel)
> row2opts(name("Group 2") msymbol(T) nolabel)

Biplot of 5 observations and 3 variables
  Explained variance by component 1 = 0.6283
  Explained variance by component 2 = 0.3717
  Total explained variance = 1.0000
```



In the above example, groups are defined by a single variable group but you can specify multiple variables with `rowover()`. The rendition of group markers is controlled by options `row1opts()` and `row2opts()`. The marker labels are disabled by using the `no1label` option.



► Example 2

Table 7.1 of [Cox and Cox \(2001\)](#) provides the scores of 10 Renaissance painters on four attributes using a scale from 0 to 20, as judged by Roger de Piles in the 17th century.

```
. use https://www.stata-press.com/data/r18/renpainters, clear
(Scores by Roger de Piles for Renaissance painters)
. list, abbrev(12)
```

	painter	composition	drawing	colour	expression
1.	Del Sarto	12	16	9	8
2.	Del Piombo	8	13	16	7
3.	Da Udine	10	8	16	3
4.	Giulio Romano	15	16	4	14
5.	Da Vinci	15	16	4	14
6.	Michelangelo	8	17	4	8
7.	Fr. Penni	0	15	8	0
8.	Perino del Vaga	15	16	7	6
9.	Perugino	4	12	10	4
10.	Raphael	17	18	12	18


```
. biplot composition-expression, alpha(1) stretch(10) table
> rowopts(name(Painters)) rowlabel(painter) colopts(name(Attributes))
> title(Renaissance painters)
```


Biplot of 10 painters and 4 attributes

```
Explained variance by component 1 = 0.6700
Explained variance by component 2 = 0.2375
Total explained variance = 0.9075
```

Biplot coordinates

Painters	dim1	dim2
Del Sarto	1.2120	0.0739
Del Piombo	-4.5003	5.7309
Da Udine	-7.2024	7.5745
Giulio Rom~o	8.4631	-2.5503
Da Vinci	8.4631	-2.5503
Michelangelo	0.1284	-5.9578
Fr Penni	-11.9449	-5.4510
Perino del~a	2.2564	-0.9193
Perugino	-7.8886	-0.8757
Raphael	11.0131	4.9251

Attributes	dim1	dim2
composition	6.4025	3.3319
drawing	2.4952	-3.3422
colour	-2.4557	8.7294
expression	6.8375	1.2348



`alpha(1)` gave us an α of 1. `stretch(10)` made the arrows 10 times longer. `table` requested that the biplot coordinate table be displayed. `rowopts()` and `colopts()` affected the rendition of the rows (observations) and columns (variables). The `name()` suboption provided a name to use instead of the default names “Observations” and “Variables” in the graph legend and in the biplot coordinate table. The `rowlabel(painter)` option requested that the variable `painter` be used to label the row points (observations) in both the graph and table. The `title()` option was used to override the default title.

The default is to produce a square graphing region. Because the x axis containing the first component has more variability than the y axis containing the second component, there are often no observations or arrows appearing in the upper and lower regions of the graph. The `autoaspect`

option sets the aspect ratio and the x -axis and y -axis scales so that more of the graph region is used while maintaining the equivalent interpretation of distance for the x and y axes.

Here is the previous biplot with the omission of the `table` option and the addition of the `autoaspect` option. We also add the `ynegate` option to invert the orientation of the data in the y -axis direction to match the orientation shown in figure 7.1 of Cox and Cox (2001). We add the `negcol` option to include column (variable) arrows pointing in the negative directions, and the rendition of these negative columns (variables) is controlled by `negcolopts()`.


```
. biplot composition-expression, autoaspect alpha(1) stretch(10) ynegate
> rowopts(name(Painters)) rowlabel(painter) colopts(name(Attributes))
> title(Renaissance painters) negcol negcolopts(name(-Attributes))
```

Biplot of 10 painters and 4 attributes

Explained variance by component 1 = 0.6700

Explained variance by component 2 = 0.2375

Total explained variance = 0.9075



4

Stored results

`biplot` stores the following in `r()`:

Scalars

`r(rho1)` explained variance by component 1
`r(rho2)` explained variance by component 2
`r(rho)` total explained variance
`r(alpha)` value of `alpha()` option

Matrices

`r(U)` biplot coordinates for the observations; stored only if the row dimension does not exceed Stata's maximum matrix size; as an alternative, use `generate()` to store biplot coordinates for the observations in variables
`r(V)` biplot coordinates for the variables
`r(Vstretch)` biplot coordinates for the variables times `stretch()` factor

Methods and formulas

Let \mathbf{X} be the centered (standardized if `std` is specified) data with N rows (observations) and p columns (variables). A biplot splits the information in \mathbf{X} into a portion related to the observations (rows of \mathbf{X}) and a portion related to the variables (columns of \mathbf{X}). This task is done using the singular value decomposition (SVD).

$$\mathbf{X} = \mathbf{U}\mathbf{\Lambda}\mathbf{V}'$$

The biplot formula is derived from this SVD by first splitting $\mathbf{\Lambda}$, a diagonal matrix, into

$$\mathbf{\Lambda} = \mathbf{\Lambda}^\alpha \mathbf{\Lambda}^{1-\alpha}$$

and then retaining the first two columns of \mathbf{U} , the first two columns of \mathbf{V} , and the first two rows and columns of $\mathbf{\Lambda}$. Using the subscript 2 to denote this, the biplot formula is

$$\mathbf{X} \approx \mathbf{U}_2 \mathbf{\Lambda}_2^\alpha \mathbf{\Lambda}_2^{1-\alpha} \mathbf{V}_2'$$

where $0 \leq \alpha \leq 1$. This is then written as

$$\mathbf{X} \approx (\mathbf{U}_2 \mathbf{\Lambda}_2^\alpha)(\mathbf{V}_2 \mathbf{\Lambda}_2^{1-\alpha})'$$

$\mathbf{U}_2 \mathbf{\Lambda}_2^\alpha$ contains the plotting coordinates corresponding to observations (rows) and $\mathbf{V}_2 \mathbf{\Lambda}_2^{1-\alpha}$ contains the plotting coordinates corresponding to variables (columns). In a biplot, the row coordinates are plotted as symbols and the column coordinates are plotted as arrows from the origin.

Let λ_i be the i th diagonal of $\mathbf{\Lambda}$. The explained variance for component 1 is

$$\rho_1 = \left\{ \sum_{i=1}^p \lambda_i^2 \right\}^{-1} \lambda_1^2$$

and for component 2 is

$$\rho_2 = \left\{ \sum_{i=1}^p \lambda_i^2 \right\}^{-1} \lambda_2^2$$

The total explained variance is

$$\rho = \rho_1 + \rho_2$$

Acknowledgment

Several biplot options were based on the work of Ulrich Kohler (2004) of Methoden der Empirischen Sozialforschung at Universität Potsdam, Germany, and coauthor of the Stata Press book *Data Analysis Using Stata*.

Kuno Ruben Gabriel (1929–2003) was born in Germany, raised in Israel, and studied at the London School of Economics and the Hebrew University of Jerusalem, earning a doctorate in demography in 1957. After several years on the faculty at the Hebrew University, he moved to the University of Rochester in 1975. His wide-ranging statistical interests spanned applications in meteorology, including weather-modification experiments, and medicine. Gabriel's best-known contribution is the biplot.

References

- Cox, T. F., and M. A. A. Cox. 2001. *Multidimensional Scaling*. 2nd ed. Boca Raton, FL: Chapman and Hall/CRC.
- Gabriel, K. R. 1971. The biplot graphical display of matrices with application to principal component analysis. *Biometrika* 58: 453–467. <https://doi.org/10.2307/2334381>.
- . 1981. Biplot display of multivariate matrices for inspection of data and diagnosis. In *Interpreting Multivariate Data*, ed. V. Barnett, 571–572. Chichester, UK: Wiley.
- Gordon, A. D. 1999. *Classification*. 2nd ed. Boca Raton, FL: Chapman and Hall/CRC.
- Gower, J. C., and D. J. Hand. 1996. *Biplots*. London: Chapman and Hall.
- Hall, W. J. 2003. Ruben Gabriel, 1929–2003. *IMS Bulletin* 32(4): 9.
- Jacoby, W. G. 1998. *Statistical Graphics for Visualizing Multivariate Data*. Thousand Oaks, CA: Sage.
- Jolliffe, I. T. 2002. *Principal Component Analysis*. 2nd ed. New York: Springer.
- Kohler, U. 2004. Biplots, revisited. 2004 German Stata Users Group meeting. <https://www.stata.com/meeting/2german/>.
- Kohler, U., and M. M. Luniak. 2005. [Data inspection using biplots](#). *Stata Journal* 5: 208–223.
- Rencher, A. C., and W. F. Christensen. 2012. *Methods of Multivariate Analysis*. 3rd ed. Hoboken, NJ: Wiley.
- Seber, G. A. F. 1984. *Multivariate Observations*. New York: Wiley.

Also see

- [MV] [ca](#) — Simple correspondence analysis
- [MV] [mds](#) — Multidimensional scaling for two-way data
- [MV] [pca](#) — Principal component analysis

Stata, Stata Press, and Mata are registered trademarks of StataCorp LLC. Stata and Stata Press are registered trademarks with the World Intellectual Property Organization of the United Nations. StataNow and NetCourseNow are trademarks of StataCorp LLC. Other brand and product names are registered trademarks or trademarks of their respective companies. Copyright © 1985–2023 StataCorp LLC, College Station, TX, USA. All rights reserved.



For suggested citations, see the FAQ on [citing Stata documentation](#).